Computing Line Integrals

Worksheet by Mike May, S.J. - maymk@slu.edu

Revised by Russell Blyth - blythrd@slu.edu

```
> restart:
```

Maple can be used to set up and evaluate line integrals over parameterized curves. We will walk through a step by step procedure, then produce new procedures that do everything in one command.

A step by step example

First we define a vector field that we want to integrate, the parameterized path we are integrating over, and the limits on the parameter. For our example we will use a familiar circular vector field and integrate over half an ellipse.

```
> vfield := [-y,x];
  path := [3*cos(2*t), 2*sin(2*t)];
  trange := t=0..Pi/2;
```

$$vfield := [-y, x]$$

$$path := [3\cos(2t), 2\sin(2t)]$$

$$trange := t = 0 .. \frac{1}{2}\pi$$

The process of integrating the line integral over the parameterized curve can be broken into 6 steps:

1) Plot a graph of the vector field and the parameterized curve.

2) Set up the line integral with the integrand equal to the dot product of the vector field and the derivative of the parameterized path.

3) Substitute the parameterization of the path into the field to make the field a vector valued function of the parameter.

4) Take the derivative of the path with respect to the parameter.

5) Evaluate the dot product and simplify. This reduces the problem to an integral over an interval.

6) Evaluate the integral.

We are ready to walk through the steps one at a time. As we work through the example with Maple, you should make sure that you can also do each step by hand.

The first step is to graph the vector field and the parameterized curve.

```
> vfieldplot := plots[fieldplot](vfield,x=-5..5,y=-5..5, grid=[7,
  7]):
  parampath := plot([op(path),trange],thickness=2):
  plots[display]({vfieldplot, parampath});
```



The next step is to set up the line integral

```
> Int((vfield*Diff(path,t)), trange);
```

$$\int_0^{\frac{1}{2}\pi} ([-y, x])\ (Diff(3\cos(2t), 2\sin(2t), t))\, dt$$

The third step is to replace x and y and the vector field with the parameterizations of x and y at the appropriate point on the path.

```
> paramfield := [subs(x=path[1],y=path[2], vfield[1]),
                 subs(x=path[1],y=path[2], vfield[2])]:
  print(`the function on the path is `, paramfield);
  print(Int((paramfield*Diff(path,t)), trange));
```

the function on the path is , $[-2\sin(2t), 3\cos(2t)]$

$$\int_0^{\frac{1}{2}\pi} ([-2\sin(2t), 3\cos(2t)])\ (Diff(3\cos(2t), 2\sin(2t), t))\, dt$$

The fourth step is to take the derivative of the parameterized path.

```
> dpath := diff(path, t):
  print(`the derivative of the path is `, diff(path, t));
  print(`I = `,Int((paramfield*dpath), trange));
```

$$the\ derivative\ of\ the\ path\ is,\ [-6\sin(2t), 4\cos(2t)]$$

$$I =, \int_0^{\frac{1}{2}\pi} ([-2\sin(2t), 3\cos(2t)])\ ([-6\sin(2t), 4\cos(2t)])\, dt$$

The fifth step is to evaluate the dot product in the integrand and simplify.

```
> integrand := simplify(ListTools[DotProduct](paramfield,dpath));
  print(`The integrand is `, integrand);
  print(`I = `,Int(integrand, trange));
```

$$integrand := 12$$

$$The\ integrand\ is,\ 12$$

$$I =, \int_0^{\frac{1}{2}\pi} 12\, dt$$

We now evaluate the integral and evaluate the result numerically if needed.

```
> intval := int(integrand, trange);
  evalf(intval);
```

$$intval := 6\pi$$

$$18.84955592$$

```
>
```

An automated approach

For convenience we block the code into two procedures we can use, one for plotting, and one for setting up the integral and evaluating it. This allows us to modify the problem with a minimal amount of work.

```
> pathplot := proc(vecfield, path, trange, xrange, yrange)
     local vfieldplot, parampath;
     vfieldplot := plots[fieldplot](vecfield,xrange,yrange, grid=
  [7,7]):
     parampath := plot([op(path),trange],thickness=2):
     plots[display]({vfieldplot, parampath});
  end:
  lineintegral:= proc(vecfield, path, trange)
     local intval, paramfield, dpath, integrand;
     print(`the vector field is `, vecfield);
     print(`the path is `, path, ` with `, trange);
     print(Int((vecfield*Diff(path,t)), trange));
     paramfield := [subs(x=path[1],y=path[2], vecfield[1]),
                    subs(x=path[1],y=path[2], vecfield[2])];
     print(`the funtion on the path is `, paramfield);
     print(Int((paramfield*Diff(path,t)), trange));
     dpath := diff(path, t);
     print(`the derivative of the path is `, diff(path, t));
     print(Int((paramfield*dpath), trange));
     integrand := simplify(ListTools[DotProduct](paramfield,
  dpath));
     print(`The integrand is `, integrand);
     print(`I = `,Int(integrand, trange));
     intval := int(integrand, trange);
     print(`the integral is `, intval);
     print(evalf(intval));
  end:
```

Thus, we can get the results by entering the vectorfield and path and executing the two commands.

```
> vfield := [-y,x];
  path := [1+3*cos(t), -4+2*sin(t)];
  trange := t=0..Pi;
  pathplot(vfield, path, trange, x=-5..5, y=-5..5);
  lineintegral(vfield, path, trange);
```

$$vfield := [-y, x]$$

$$path := [1+3\cos(t), -4+2\sin(t)]$$

$$trange := t = 0..\pi$$



$$\textit{the vector field is}, [-y, x]$$

$$\textit{the path is}, [1+3\cos(t), -4+2\sin(t)], \textit{ with}, t = 0..\pi$$

$$\int_0^{\pi} ([-y, x])\ (Diff(1+3\cos(t), -4+2\sin(t), t))\, dt$$

$$\textit{the funtion on the path is}, [4-2\sin(t), 1+3\cos(t)]$$

$$\int_0^{\pi} ([4-2\sin(t), 1+3\cos(t)])\ (Diff(1+3\cos(t), -4+2\sin(t), t))\, dt$$

$$\textit{the derivative of the path is}, [-3\sin(t), 2\cos(t)]$$

$$\int_0^{\pi} ([4-2\sin(t), 1+3\cos(t)])\,([-3\sin(t), 2\cos(t)])\,dt$$

$$\textit{The integrand is}, -12\sin(t)+2\cos(t)+6$$

$$I = , \int_0^{\pi} (-12\sin(t)+2\cos(t)+6)\,dt$$

$$\textit{the integral is}, -24+6\pi$$

$$-5.15044408$$

>

Line Integrals in 3-D

If the vector field is in three dimensions we need to slightly modify the two commands.

```
> pathplot3d := proc(vecfield, path, trange, xrange, yrange, 
zrange)
    local vfieldplot, parampath;
    vfieldplot := plots[fieldplot3d](vecfield,xrange,yrange,
zrange, grid=[7,7,7], color=blue):
    parampath := plots[spacecurve]([op(path),trange],thickness=
2, color=red):
    plots[display]({vfieldplot, parampath});
end:

lineintegral3d:= proc(vecfield, path, trange)
    local intval, paramfield, dpath, integrand;
    print(`the vector field `, vecfield);
    print(`the path `, path, ` with `, trange);
    print(Int((vecfield*Diff(path,t)), trange));
    paramfield := [subs(x=path[1],y=path[2],z=path[3], vecfield
[1]),
                  subs(x=path[1],y=path[2],z=path[3], vecfield
[2]),
                  subs(x=path[1],y=path[2],z=path[3], vecfield
[3])];
    print(`the funtion on the path is `, paramfield);
    print(Int((paramfield*Diff(path,t)), trange));
    dpath := diff(path, t);
    print(`the derivative of the path is `, diff(path, t));
    print(Int((paramfield*dpath), trange));
    integrand := simplify(ListTools[DotProduct](paramfield,
dpath));
    print(`The integrand is `, integrand);
    print(`I = `,Int(integrand, trange));
    intval := int(integrand, trange);
    print(`the integral is `, intval);
    print(evalf(intval));
end:
```

Using this command we consider the following example.

```
> vfield := [x, z, -x*y];
path := [cos(t), sin(t), 2*t];
trange := t=0..3*Pi;
pathplot3d(vfield,path,trange,x=-2..2,y=-2..2,z=0..19);
lineintegral3d(vfield, path, trange);
```

$$vfield := [x, z, -x\,y]$$

$$path := [\cos(t), \sin(t), 2\,t]$$

$trange := t = 0 .. 3\pi$



the vector field, $[x, z, -x\,y]$

the path, $[\cos(t), \sin(t), 2\,t]$, *with*, $t = 0 .. 3\pi$

$$\int_0^{3\pi} ([x, z, -x\,y])\ (Diff(\cos(t), \sin(t), 2\,t, t))\ \mathrm{d}t$$

the funtion on the path is, $[\cos(t), 2\,t, -\cos(t)\sin(t)]$

$$\int_0^{3\pi} ([\cos(t), 2\,t, -\cos(t)\sin(t)])\ (Diff(\cos(t), \sin(t), 2\,t, t))\ \mathrm{d}t$$

the derivative of the path is, $[-\sin(t), \cos(t), 2]$

$$\int_0^{3\pi} ([\cos(t), 2\,t, -\cos(t)\sin(t)])\ ([-\sin(t), \cos(t), 2])\ \mathrm{d}t$$

The integrand is, $\cos(t)\ (-3\sin(t) + 2\,t)$

$$I = , \int_0^{3\pi} \cos(t)\ (-3\sin(t) + 2\,t)\ \mathrm{d}t$$

the integral is , −4

−4.

>

Exercises

1. Evaluate the line integral where F = [ln(y), ln(x)] and C is the curve given parametrically by [2t, t^2] with t=2..4.

>

2) Evaluate the line integral where F = [2y, -sin(y)] and C is the unit circle traced in a counterclockwise direction from the point (1, 0).

>

3) Evaluate the line integral where F = e^x i + e^y j and C is the part of the ellipse x^2 + 4y^2 = 4 from the point (0, 1) to the point (2, 0) traversed in a clockwise direction.

>

4) Evaluate the line integral where F = -y i + x j + 5 k and C is the helix [cos(t), sin(t) t] for t=0..4Pi.

>

>